

FÜGGVÉNYKÖNYVTÁRAK:


Tartalmazza az összes függvényt és az egyes függvények használatának rövid leírását

KÉPLETVIZSGÁLAT:

Elődök / Utódok mutatása megmutatja mely cellák befolyásolják, vagy befolyásolták a kiválasztott cella értékét.

Képletek a beírt képleteket jeleníti meg, nem a képlet eredményét

Képletkiértékelő lehetővé teszi hogy belépjen a képletbe és lépésről lépésre hajtsa azt végre



Az összes Excel képlet “=” jellel kezdődik, amit a függvény vagy egyszerűbb esetben a cellahivatkozások követnek, amelyekkel a műveletet végezzük

Amint elkezd beírni a képletet, megjelenik egy felbukkanó ablak, amely végigvezet minden lépésen

A1

Egycellás hivatkozás egy cella helye a munkalapon, egy oszlop (A – XFD) és egy sor (1–1048 576) metszéspontja

	A	B	C
1			
2			
3			
4			
5			

A1:C4

Cellatartomány szomszédos cellák csoportja, a bal felső (A1) és a jobb alsó (C4) cellák elhelyezkedése alapján, elválasztva egy „:” karakterrel

	A	B	C
1			
2			
3			
4			
5			

A1,C4

Nem szomszédos cellahivatkozás nem szomszédos cellák, vagy cellatartományok csoportja, elválasztva “,” karakterrel

	A	B	C
1			
2			
3			
4			
5			

Miért tartalmaz néhány cellahivatkozás „\$” jelet?

A \$ jeleket **Rögzített, Relatív** vagy **Vegyes** cellahivatkozások létrehozására használjuk;
a \$ lezár egy adott cellatartományt vagy cellahivatkozást, hogy az ne változzon, ha a képletet más cellákba másoljuk

Példák:

\$A\$1 = Rögzített oszlop, Rögzített sor

A\$1 = Relatív oszlop, Rögzített sor


\$A1 = Rögzített oszlop, Relatív sor

A1 = Relatív oszlop, Relatív sor


TIPP:

Válassza ki a képlet egy részét a kurzorral és használja az „F4” billentyűt a referencia típusok gyors görgetéséhez.


	A	B	C
1	\$A\$1		
2			
3			
4			\$A\$1
5			




	A	B	C
1	A\$1		
2			
3			
4			C\$1
5			



	A	B	C
1	\$A1		
2			
3			
4			\$A4
5			



	A	B	C
1	A1		
2			
3			
4			C4
5			



Hiba típusa	Mit jelent?	Hogyan orvosoljuk?
#####	<i>Az oszlopszélesség nem teszi lehetővé a cella tartalmának megjelenítését</i>	<i>Állítsa át az oszlop szélességét</i>
#NÉV?	<i>Ismeretlen függvény, cellahivatkozás van a képletben</i>	<i>Ellenőrizze, hogy a használt függvény neve helyes, hogy a hivatkozások érvényesek és helyesen vannak megadva</i>
#ÉRTÉK!	<i>A képlet nem megfelelő típusú értéket tartalmaz</i>	<i>Ellenőrizze, hogy a képlet nem próbál-e számtani műveletet végrehajtani szöveges értékkel</i>
#ZÉROOSZTÓ!	<i>A képlet 0-val vagy üres cellával oszt</i>	<i>Ellenőrizze az osztó értékét</i>
#HIV	<i>Érvénytelen cellára hivatkozás</i>	<i>Győződjön meg arról, hogy nem mozgatta, törölte vagy cserélte a képletében hivatkozott cellákat</i>

AZ **F4** billentyűnek két hasznos funkciója van:

1) Módosítja a cellahivatkozás típusát

Ha a kurzorral kiválasztunk egy cellahivatkozást a képleten belül, az F4 gomb a rögzített, relatív és abszolút referenciatípusokon keresztül fog ciklust futtatni



2) Megismétli az utolsó parancsot

Az F4 megismétli az utolsó felhasználói műveletet, például sorok vagy oszlopok beszúrását / törlését, formázást

Megjegyzés: Az F4 nem fogja megismételni a megadott értékeket vagy képleteket

Az **F2** billentyű megjeleníti újra a képletet

TIPP:

Az F2 segítséget nyújthat a képlethibák diagnosztizálásában, vagy a cellahivatkozások gyors módosításában

A **CTRL** funkció kombinálható különféle billentyűkkel, például:

1) CTRL- NYÍLAK

Balra, jobbra, felső vagy alsó szélre ugrik egy adattáblában

2) CTRL-SHIFT-NYÍLAK

Mint fenn, csak ki is jelöli a cellákat

3) CTRL-PAGE UP/DOWN

A munkalapok között lépkedhetünk

CTRL-SHIFT-JOBBRA NYÍL

	A	B	C	D	E	F	G	H
1	77	847	482	847	916	329	796	
2	183	852	286	275	177	476	224	
3	252	117	134	865	242	822	705	
4	711	507	125	910	348	529	491	
5	842	12	837	491	221	595	369	
6	782	39	906	245	286	753	964	
7	820	678	473	777	172	655	984	
8	321	164	803	461	225	560	652	
9	374	447	395	232	742	101	916	
10	891	966	861	898	719	757	141	
11	718	775	635	817	550	703	602	
12								

CTRL-SHIFT- LE NYÍL

	A	B	C	D	E	F	G	H
1	77	847	482	847	916	329	796	
2	183	852	286	275	177	476	224	
3	252	117	134	865	242	822	705	
4	711	507	125	910	348	529	491	
5	842	12	837	491	221	595	369	
6	782	39	906	245	286	753	964	
7	820	678	473	777	172	655	984	
8	321	164	803	461	225	560	652	
9	374	447	395	232	742	101	916	
10	891	966	861	898	719	757	141	
11	718	775	635	817	550	703	602	
12								

Alapvető számítások elvégzéséhez a következő függvényeket használjuk:

	A	B	C	D
1	Értékek		Számok	
2	90		Darabszáma	17
3	13			
4	22		Átlaga	53,71
5	98			
6	61		Mediánja	56
7	69			
8	50		Módusza	22
9	91			
10	16		Maximuma	98
11	23			
12	60		Minimuma	13
13	22			
14	56		Összege	913
15	54			
16	87		75 százaléka	69
17	33			
18	68		Szórása	28,27

=DARAB(A2:A18)

=ÁTLAG(A2:A18)

=MEDIÁN(A2:A18)

=MÓDUSZ(A2:A18)

=MAX(A2:A18)

=MIN(A2:A18)

=SZUM(A2:A18)

=PERCENTILIS(A2:A18;0,75)

=SZÓRÁS(A2:A18)

	A
1	Értékek
2	90
3	13
4	22
5	98
6	61
7	69
8	50

SORSZÁM(A2;A2:A8) = 2

SORSZÁM(A3;A2:A8) = 7 (legkisebb)

SORSZÁM(A4;A2:A8) = 6

SORSZÁM(A5;A2:A8) = 1 (legnagyobb)

SORSZÁM(A6;A2:A8) = 4

SORSZÁM(A7;A2:A8) = 3

SORSZÁM(A8;A2:A8) = 5

A **KICS/NAGY** függvények egy tömb
n-edik legkisebb/legnagyobb
értékét adják vissza

A **SORSZÁM** függvény egy adott
érték helyét adja meg egy
tartományon belül

	A
1	Értékek
2	90
3	13
4	22
5	98
6	61
7	69
8	50

NAGY(A2:A8;2) = 90

(a második legnagyobb érték a 90)

KICSI(A2:A8;3) = 50

(a harmadik legkisebb érték az 50)

	A
1	Értékek
2	90
3	13
4	22
5	98
6	61
7	69
8	50
9	91
10	16
11	23
12	60
13	22
14	56
15	54
16	87
17	33
18	68

A **PERCENTILIS függvény** megadja melyik az szám, amelynél egy tömb meghatározott százaléku eleme kisebb vagy egyenlő

=PERCENTILIS(A2:A18; x)

Milyen adattartományban keresünk?

Melyik elemre.
Egy 0 és 1 közötti szám

PERCENTILIS(A2:A18;0) = 13
(*legkisebb*)

PERCENTILIS(A2:A18;0,5) = 56
(*medián*)

A **VÉL()** és a **VÉLETLEN.KÖZÖTT** függvények alkalmasak véletlenszámok generálására:

	A	B	C	D	E
1	0.5173	0.4091	0.7560	0.9012	0.2167
2	0.0906	0.2317	0.0906	0.5856	0.8646
3	0.1544	0.8240	0.4279	0.8782	0.7795
4	0.0097	0.0872	0.7740	0.9137	0.7815
5	0.2089	0.7028	0.0449	0.8173	0.9983
6	0.0761	0.4388	0.4056	0.5639	0.0668

A **VÉL()** függvény 0 és 1 közötti számokat állít elő (15 tizedessel)

A **VÉLETLEN.KÖZÖTT** függvény egész számokat állít elő a megadott két érték között

	A	B	C	D	E
1	83	23	64	62	92
2	59	45	40	50	91
3	24	37	70	30	32
4	54	85	69	55	3
5	73	12	36	53	2
6	29	72	68	59	99

=VÉLETLEN.KÖZÖTT(0;100)

A **DARABTELI**, **SZUMHA** és az **ÁTLAGHA** függvények csak bizonyos kritériumnak megfelelő cellákat vesznek figyelembe

	A	B
1	Név	Kor
2	Gábor	90
3	Márton	13
4	Róbert	22
5	Tamás	98
6	Gábor	61
7	Tamás	68
8	Tamás	50
9	Márton	91
10	Gábor	16
11	Márton	23
12	Tamás	60
13	Róbert	22
14	Márton	56
15	Gábor	54
16	Gábor	87
17	Róbert	33
18	Róbert	68
19	Róbert	45
20	Gábor	21

=DARABTELI(**tartomány**; kritérium)

=SZUMHA(**tartomány**; kritérium; **összeztartomány**)

=ÁTLAGHA(**tartomány**; kritérium; **átlagtartomány**)

Mely celláknak kell
megfelelnie a
kritériumoknak?

Körülmények, kritériumok
leírása

Hol vannak azon
értékek, amiket
összegezni, vagy
átlagolni kell?

DARABTELI(B2:B20;22) = **2** (22 évesek száma)

SZUMHA(A2:A20;"Róbert";B2:B20) = **190** (Róbertek összeadott életkora)

SZUMHA(A2:A20;"<>Tamás";B2:B20) = **702** (Mindenki összeadott életkora, kivéve Tamásét)

ÁTLAGHA(A2:A20;"Márton";B2:B20) = **45,75** (Mártonok átlagéletkora)

A **DARABHATÖBB**, **SZUMHATÖBB** és az **ÁTLAGHATÖBB** az előző függvények továbbfejlesztése, ahol több tartomány és a hozzá tartozó kritériumok megadhatók

=DARABHATÖBB(tartomány1; kritérium1; tartomány2; kritérium2...)

=SZUMHATÖBB(összegtartomány; tartomány1; kritérium1; tartomány2; kritérium2...)

=ÁTLAGHATÖBB(átlagtartomány; tartomány1; kritérium1; tartomány2; kritérium2...)

	A	B	C	D
1	Hónap	Taktika	Kampány	Kattintások
2	Jan	Keresés	Google	166
3	Jan	Keresés	MSN	263
4	Jan	Megjelenés	Szöveges	289
5	Jan	Megjelenés	Célzott	137
6	Feb	Keresés	Google	124
7	Feb	Keresés	MSN	311
8	Feb	Megjelenés	Szöveges	350
9	Feb	Megjelenés	Célzott	384
10	Mar	Keresés	Google	168
11	Mar	Keresés	MSN	358
12	Mar	Megjelenés	Szöveges	347
13	Mar	Megjelenés	Célzott	390

DARABHATÖBB(B2:B13; "Keresés"; D2:D13; ">200") = 3

(Azon reklámok száma, ahol a Taktika Keresés és Kattintások száma több mint 200)

SZUMHATÖBB(D2:D13; A2:A13; "Feb"; B2:B13; "Megjelenés") = 734

(Azon kattintások összege, ahol a Hónap Feb(ruár) és a Taktika Megjelenés)

ÁTLAGHATÖBB(D2:D13; A2:A13; "Jan"; C2:C13; "MSN") = 263

(Azon kattintások átlaga, ahol a Hónap Jan(uár) és a Kampány az MSN)

Az Excel minden dátumához egy társított dátumérték tartozik, azaz az Excel az idő múlásával tartja számon a dátumokat

Az Excel felismeri a legtöbb beírt dátumot, és automatikusan alkalmaz egy általános formátumot (év.hónap.nap) és társítja hozzá a dátumértéket

Megjegyzés: Ha olyan dátumot írunk be, amelyet az Excel NEM ismer fel, akkor szövegként kezeli és hozzá tartozó dátumértéke sem lesz

	A	B
1	Dátum	Dátumérték
2	1900.01.01	1
3	1949.01.21	17919
4	1976.10.28	28061
5	2020.03.01	43891
6	2020.03.01 12:00	43891,5
7	2020.03.01 18:00	43891,75
8	9999.12.31	2958465


1900 január 1 az első nap, amelynek hozzárendelt dátumértéke van, mégpedig az 1. 1976 október 28-a a 28061. nap 1900 január elseje óta, tehát a dátumértéke 28061.

A dátumértékek a napok töredékeit is jelzik: 43891,5 azt jelzi, hogy 2020/03/01 dele (50%-a az adott napnak)


A jövőbe is tekinthetünk, egészen 9999.12.31-ig. Onnantól már nem férne el négy karakteren az évszám, így tovább nem tudja kezelni az Excel.

A dátumok formázásához választhatunk előre beállított opciók közül a „**Cella formázása**” párbeszédpanelen, vagy készíthetünk saját **egyéni formátumot**

Előre beállított formátumok:




Egyéni formátumok:



*Készíthet saját egyéni formátumokat a dátum- és idő kódok kombinációival.
Néhány példa:*

éé = az év utolsó két karaktere (15)
éééé = teljes évszám (2015)
h = hónap, vezető nulla nélkül (1-12)
hh = hónap, vezető nullával (01-12)
hhh = hónap neve röviden (Jan-Dec)
hhhh = hónap neve (Január-December)
n = nap, vezető nulla nélkül (1-31)
nn = nap, vezető nullával (01-31)
nnn = a hét napja röviden (H-V)
nnnn = a hét napja (Hétfő-Vasárnap)

Amikor egy dátumot tartalmazó cella sarkát húzzuk, az Excel automatikusan **nap** értékeket lép, ha a másodlagos egérgombbal alkalmazzuk a gyorskitöltést, akkor választhatunk.



Cellák másolása = Megismétli ugyanazt az értéket minden cellában

Kitöltés napokkal = A dátumot cellánként 1 nappal növeli

Kitöltés hétköznapokkal = A dátumot cellánként 1 nappal növeli (kivéve a hétvégeket)

Kitöltés hónapokkal = A dátumot cellánként 1 hónappal növeli

Kitöltés évekkel = A dátumot cellánként 1 évvel növeli

A **MA()** és a **MOST()** függvény az aktuális dátumot és a pontos időt adja vissza.

Megjegyzés: Ezek frissülő függvények, vagyis minden munkalap számításnál megváltoznak

MA() =	2020.03.01
MOST() =	2020.03.01 14:36

Az **ÉV**, **HÓNAP**, **NAP**, **ÓRA**, **PERCEK** és **MPERC** függvények kibontják egy adott dátum egyes összetevőit:

	ÉV	HÓNAP	NAP	ÓRA	PERC	MÁSODPERC
2020.03.01 14:46	2020	3	1	14	46	38
	=ÉV(B19)	=HÓNAP(B19)	=NAP(B19)	=ÓRA(B19)	=PERCEK(B19)	=MPERC(B19)

Használhatjuk a **HÓNAP.UTOLSÓ.NAP** függvényt az adott hónap utolsó napjának kiszámításához, vagy az előző vagy a jövőbeli hónapok kezdő / befejező dátumának kiszámításához

=HÓNAP.UTOLSÓ.NAP(start_dátum, hónapok)

Hivatkozás egy cellára vagy
egy dátum érték

A kezdő / aktuális dátum előtti vagy utáni hónapok száma (a pozitív
szám a jövőbeli dátumot eredményez, a negatív szám a múlt dátumát
adja meg)

	A	B	C
1			
2		Aktuális dátum:	2020.02.15
3			
4		Hónap vége:	2020.02.29
5		Hónap eleje:	2020.02.01
6		Következő hónap eleje:	2020.03.01

→ =HÓNAP.UTOLSÓ.NAP(C2, 0)


→ =HÓNAP.UTOLSÓ.NAP(C2, -1)+1

→ =HÓNAP.UTOLSÓ.NAP(C2, 0)+1


Ha azt szeretnénk tudni, hogy egy adott nap a hét melyik napjára esik, akkor kétféleképpen tehetjük meg:

- 1) Használjuk az egyéni cellaformázást, "nnn" (H) vagy "nnnn" (hétfő)
- Ez viszont nem változtatja meg a cella értékét, csak az érték megjelenítését
- 2) Használjuk a **HÉT.NAPJA** függvényt, ami a hét egy adott napjának megfelelő értéket ad vissza (1-7 vagy 0-6)

=HÉT.NAPJA(dátumérték, [visszatérés_típusa])



Ez egy dátumot vagy időt tartalmazó cella, vagy érték



0 (alapértelmezett) = Vasárnap (1) ... Szombat (7)

1 = Hétfő (1) ... Vasárnap (7)

3 = Hétfő (0) ... Vasárnap (6)

A szöveges függvények felhasználhatók a formázás szabványosítására, különös tekintettel a **KIMETSZ**, **KISBETŰ**, **NAGYBETŰS** és **TNÉV** függvényeket

Példaszöveg	Képlet	Eredmény	Leírás
Géza kék az ég	=KIMETSZ(A2)	Géza kék az ég	Eltávolítja a felesleges szóközöket
Géza kék az ég	=KISBETŰ(A3)	géza kék az ég	Minden karaktert kisbetűsre alakít
Géza kék az ég	=NAGYBETŰS(A4)	GÉZA KÉK AZ ÉG	Minden karaktert nagybetűsre alakít
Géza kék az ég	=TNÉV(A5)	Géza Kék Az Ég	A szókezdő betűket nagybetűsre alakítja

TIPP:

Ha két szöveges karakterlánc látszólag azonos, lehet, hogy csak annak tűnnek. Ha az egyiknek van egy szóköz a végén, akkor pontosan ugyanolyannak tűnnek, de az Excel teljesen eltérő értékeként kezeli őket; A TRIM egyenértékűvé teszi őket

Az **ÖSSZEFÜZ** függvény lehetővé teszi, hogy szövegek és cellák értékeit vagy képlet kimeneteket egyetlen szöveges karakterlánccá egyesítsük

Megjegyzés: Az **"=ÖSSZEFÜZ(Szöveg1, Szöveg2...)"** függvény helyett használható a **„&”** karakter.

Vezetéknév	Keresztnév	Képlet	Kimenet
Kormos	Attila	=A2&B2	KormosAttila
Kormos	Attila	=A3&" "&B3	Kormos Attila
Kormos	Attila	=A4&" "&BAL(B4;1)&"."	Kormos A.
Kormos	Attila	=BAL(A5;1)&"."&B5	K. Attila
Kormos	Attila	=BAL(A6;1)&"."&BAL(B6;1)&"."	K. A.

A **BAL**, **KÖZÉP** és a **JOBB** függvény adott számú karaktert ad vissza egy szöveges karakterláncból, a függvény nevéből adódóan az adott helyről. A **HOSSZ** függvény pedig a karakterlánc hosszát adja vissza.

=BAL(szöveg, [karakterek_száma])

=JOBB(szöveg, [karakterek_száma])

=KÖZÉP(szöveg, kezdő_karakter, karakterek_száma)

	A	B	C	D
1	Példaszöveg	Képlet	Kimenet	Leírás
2	MA-02215%AAA%_100	=BAL(A2;2)	MA	Visszaad két karaktert bal oldalról
3	MA-02215%AAA%_100	=KÖZÉP(A3;4;5)	02215	Visszaad öt karakter a szöveg közepéről, a negyedik karaktertől indulva
4	MA-02215%AAA%_100	=JOBB(A4;3)	100	Visszaad három karakter jobb oldalról
5	MA-02215%AAA%_100	=HOSSZ(A5)	17	Visszaadja a szöveg hosszát

A **SZÖVEG** függvény egy numerikus értéket szöveggé konvertál és egy adott formátumot rendel hozzá

=SZÖVEG(érték; egyéni_formátum)

Numerikus érték, számértékre értékelődő képlet, vagy hivatkozás egy számot tartalmazó cellára

Numerikus egyéni formátum idézőjelekkel bezárva (pl. "éééé/hh/nn", "0,00 Ft" vagy "# ##0,00")


	A	B
1	Név	Kereset
2	Tamás	210 000 Ft
3	Gábor	225 000 Ft
4	László	178 000 Ft

=A4&" keresete "&B4 →
"László keresete 178000"


=A4&" keresete "&SZÖVEG(B4;"# ### Ft") →
"László keresete 178 000 Ft"

A **SZÖVEG.KERES** függvény egy számként adja vissza a keresett szöveg első előfordulásának kezdőpozícióját egy másik szövegben. Ha nem található a keresett szöveg, **#ÉRTÉK!** hibaüzenetet kapunk


=SZÖVEG.KERES(keresett_szöveg, mely_szövegben, [kezdőérték])



Milyen karaktert vagy karakterláncot keressen?



Mely szövegben keressen?



Keressen az elejétől (alapértelmezés) vagy egy bizonyos számú karakter után?

Példaszöveg	Képlet	Kimenet	Leírás
MA-02215%AAA%_100	=SZÖVEG.KERES("%";A2)	9	% jel keresése a szöveg elejétől
MA-02215%AAA%_100	=SZÖVEG.KERES("%";A3;10)	13	% jel keresése a szöveg 10. karakterétől
MA-02215%AAA%_100	=KÖZÉP(A4;SZÖVEG.KERES("%";A4);5)	%AAA%	5 karaktert ad vissza a szöveg közepéből a % jeltől kezdődően
MA-02215%AAA%_100	=KÖZÉP(A5;SZÖVEG.KERES("%";A5)+1;3)	AAA	3 karaktert ad vissza a szöveg közepétől a % jelet követően

A **HA(SZÁM(SZÖVEG.KERES** függvények kombinációja felhasználható adatok osztályozására olyan értékek alapján, amelyek meghatározott szövegrészeket tartalmaznak

=HA(SZÁM(SZÖVEG.KERES(keresett; miben)); érték_ha_igaz; érték_ha_hamis)

Egy adott karakterláncra keres
egy adott cellában

Az egyik értéket adja vissza, ha a
karakterlánc megtalálható (TRUE), ha
nem található (FALSE)

	A	B
1	Reklámkampány	Típus
2	12983-Aff-160x90_small	Egyéb
3	982308-Disp-160x90_large	Képernyő
4	21314-Aff-160x90_small	Egyéb
5	463-Disp-160x90_small	Képernyő
6	390238-Agg-160x90_large	Egyéb

=HA(SZÁM(SZÖVEG.KERES("Disp";A2));"Képernyő";"Egyéb")

Skeresse meg az A oszlop értékeiben a „Disp” szöveget. Ha tartalmazza, akkor a B oszlopban a „Képernyő” szöveg jelenjen meg, egyébként az „Egyéb”



Az Excel összes **Logikai Operátora** egyszerű
“HA / AKKOR” kifejezésen alapul:

- HA esik, **AKKOR** viszek esernyőt*
- HA süt a nap, **AKKOR** viszek napszemüveget*
- HA süt a nap **ÉS** nyár van, **AKKOR** szabira megyek és irány a strand*

Alapvetően a működése: ***Ha az állítás igaz, akkor adjon vissza egy értéket. Egyébként adjon vissza valami mást.***

=HA(logikai_kifejezés, [Érték ha igaz], [Érték ha hamis])

Olyan kifejezés, ami csak **IGAZ** vagy **HAMIS** értéket adhat vissza

(pl. A1=„Termék” A2>10, A3<=5)

Visszaadott érték, ha a logikai kifejezés értéke **IGAZ**

Visszaadott érték, ha a logikai kifejezés értéke **HAMIS**

	A	B	C
1	Hely	Hőmérséklet	Fagy
2	Eger	-2	Igen
3	Kassa	-4	Igen
4	Pest	3	Nem
5	Tata	4	Nem
6	Pécel	2	Nem
7	Győr	-1	Igen

= HA(B2<=0;„Igen”;„Nem”)

A Fagy oszlopot „Igen” kategóriába soroljuk, ha a hőmérséklet 0 fok vagy annál alacsonyabb, egyébként „Nem”

Beágyazott HA függvények használatával több logikai kifejezést is vizsgálhatunk:

	A	B	C
1	Hely	Hőmérséklet	Klíma
2	Madrid	31	kánikula
3	Helsinki	0	fagy
4	Moszkva	11	enyhe
5	Róma	24	enyhe
6	Barrow	-5	fagy
7	Szahara	35	kánikula

= HA(B2<=0;"FAGY";HA(B2>30;"KÁNIKULA";"ENYHE"))

Ha a hőmérséklet<=0, akkor a klíma „fagy”, ha a hőmérséklet>30, akkor a klíma „kánikula”, minden egyéb esetben a klíma „enyhe”

Az Excel **ÉS** és **VAGY** utasításai lehetővé teszik, hogy egyszerre több logikai kifejezést is vizsgáljunk:

	A	B	C	D	E	F
1	Hely	Hőmérséklet	Klíma	Csapadék (mm)	Csapadék típusa	Körülmények
2	Madrid	31	kánikula	0	nincs	száraz
3	Helsinki	0	fagy	2,3	hó	nedves
4	Moszkva	11	enyhe	1,2	eső	nedves
5	Róma	24	enyhe	0,5	eső	nedves
6	Barrow	-5	fagy	1,4	hó	nedves
7	Szahara	35	kánikula	0	nincs	száraz

=HA(VAGY(E2="hó";E2="eső");"nedves";"száraz")

*Itt a körülményeket kategorizáljuk. Nedvesként, ha a csapadék típusa hó, **vagy** eső, minden egyéb esetben szárazként*

=HA(ÉS(C2="fagy";D2>0);"hó";HA(D2>0;"eső";"nincs"))

*Itt pedig a csapadék típusát. Hóként, ha a klíma fagyos **és** a csapadék mennyisége több mint 0. Esőként, ha az első logikai kifejezés nem teljesült, de a csapadék mennyisége több mint 0, minden egyéb esetben nem volt csapadék, amint nincs-el jelzünk*

Ha azt az esetet akarjuk vizsgálni, ahol a logikai kifejezés nem igaz, akkor használhatjuk a **NEM** utasítást vagy a „<>” operátort.

	A	B	C	D	E	F
1	Hely	Hőmérséklet	Klíma	Csapadék (mm)	Csapadék típusa	Körülmények
2	Madrid	31	kánikula	0	nincs	száraz
3	Helsinki	0	fagy	2,3	hó	nedves
4	Moszkva	11	enyhe	1,2	eső	nedves
5	Róma	24	enyhe	0,5	eső	nedves
6	Barrow	-5	fagy	1,4	hó	nedves
7	Szahara	35	kánikula	0	nincs	száraz


→ =HA(NEM(D2=0);"nedves";"száraz")

→ =HA(D2<>0;"nedves";"száraz")


Mindkét feltételben ugyanazt írtuk le. Ha a csapadék mennyisége nem nulla.

A **HAHIBA** függvény kiváló eszköz a nem kívánt hibaüzenetek kiküszöbölésére (**#ÉRTÉK!**, **#NÉV?**, **#ZÉRÓOSZTÓ!**, *stb.*), ami különösen fontos front end formázásnál

=HAHIBA(érték, érték_hiba_esetén)



Képlet, vagy érték ami hibát
eredményezhet



Hiba esetén
visszaadott érték

TIPP:

*Ha olyan képletet írunk, amely hibát válthat ki (például egy FKERES, ahol nem minden értékre talál egyezést), akkor írjuk fel a képletet, majd ágyazzuk bele egy **HAHIBA** függvénybe*

Az Excel a következő függvényekre IGAZ, vagy HAMIS értéket ad vissza:

ÜRES = *Ellenőrzi, hogy egy cella üres*

SZÁM = *Ellenőrzi, hogy egy cella számot tartalmaz*

SZÖVEG.E = *Ellenőrzi, hogy egy cella szöveget tartalmaz*

HIBÁS = *Ellenőrzi, hogy egy cella hibaértéket tartalmaz*

PÁROSE = *Ellenőrzi, hogy egy cella páros számot tartalmaz*

PÁRATLANE = *Ellenőrzi, hogy egy cella páratlan számot tartalmaz*

LOGIKAI = *Ellenőrzi, hogy egy cella logikai értéket tartalmaz*

KÉPLET = *Ellenőrzi, hogy egy cella képletet tartalmaz*